Shader "UShaderBook/C7/MaskTexture"
{
    Properties
    {
        _Color ("Color Tint", Color) = (1,1,1,1) // 主色调
        _MainTex ("Main Tex", 2D) = "white" {} // 主材质
        _Gloss ("Gloss", Range(8.0,256)) = 20 // 反光度
        _Specular ("Specular", Color) = (1,1,1,1) // 反光颜色
        _BumpMap ("Normal Map", 2D) = "bump" {} // 法线贴图
        _BumpScale ("Bump Scale", Float) = 1.0 // 凹凸程度
        _SpecularMask ("Specular Mask", 2D) = "white" {} // 高光遮罩
        _SpecularScale ("Specular Scale", Float) = 1.0 // 遮罩强度
    }

    SubShader
    {
        Pass{
            Tags {"LightMode"="ForwardBase"}
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #include "Lighting.cginc"
            
            fixed4 _Color;
            sampler2D _MainTex;
            float4 _MainTex_ST;
            float _Gloss;
            fixed4 _Specular;
            sampler2D _BumpMap;
            float _BumpScale;
            sampler2D _SpecularMask;
            float _SpecularScale;
            
            struct a2v{
                float4 vertex : POSITION;
                float3 normal : NORMAL;
                float4 tangent : TANGENT;
                float4 texcoord : TEXCOORD0;
            };
            
            struct v2f{
                float4 pos : SV_POSITION;
                float4 uv : TEXCOORD0;
                float3 lightDir : TEXCOORD1;
                float3 viewDir : TEXCOORD2;
            };
            
            v2f vert(a2v v){
                v2f o;
                // 裁剪空间
                o.pos = UnityObjectToClipPos(v.vertex);
                
                // 纹理坐标
                o.uv.xy = TRANSFORM_TEX(v.texcoord, _MainTex);

                TANGENT_SPACE_ROTATION;

                o.lightDir = mul(rotation, ObjSpaceLightDir(v.vertex)).xyz;
                o.viewDir = mul(rotation, ObjSpaceViewDir(v.vertex)).xyz;
                
                return o;
            }

            fixed4 frag(v2f i) : SV_TARGET{
                fixed3 tangentLightDir = normalize(i.lightDir);
                fixed3 tangentViewDir = normalize(i.viewDir);

                fixed3 tangentNormal = UnpackNormal(tex2D(_BumpMap, i.uv.xy));

                tangentNormal.xy *= _BumpScale; // 凹凸程度
                tangentNormal.z = sqrt(1.0 - saturate(dot(tangentNormal.xy, tangentNormal.xy)));

                fixed3 albedo = tex2D(_MainTex, i.uv.xy).rgb * _Color.rgb; // 世界法线
                
                // 环境光反射结果
                fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz * albedo; 

                // 漫反射结果
                fixed3 diffuse = _LightColor0.rgb * albedo * max(0, dot(tangentNormal, tangentLightDir));

                // 高光反射结果
                fixed3 halfDir = normalize(tangentLightDir + tangentViewDir);
                fixed specularMask = tex2D(_SpecularMask, i.uv).r * _SpecularScale;
                fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(max(0, dot(tangentNormal, halfDir)), _Gloss) * specularMask;

                // return fixed4(i.viewDir, 1);
                return fixed4(ambient + diffuse + specular, 1.0);
            }
            ENDCG
        }
    }
    FallBack "Specular"
}
